﻿Shader "Unlit/WhiteBlack"
{
    Properties{  

        _Diffuse("Diffuse", Color) = (1,1,1,1)  

        _OutlineCol("OutlineCol", Color) = (1,0,0,1)  

        _OutlineFactor("OutlineFactor", Range(0,1)) = 0.1  

        _MainTex("Base 2D", 2D) = "white"{}  

        _shadowColor("ShadowColor", Color) = (1,0,0,1)  		

        _shadowHeight("ShadowHeight", Range(0,1)) = 0 

        _shadowOffset("ShadowOffset", Range(0,10)) = 0 

    }  
    //子着色器    

    SubShader  

    {  

        //描边使用两个Pass，第一个pass沿法线挤出一点，只输出描边的颜色  

        Pass  

        {  

            //剔除正面，只渲染背面，对于大多数模型适用，不过如果需要背面的，就有问题了  

            Cull Front  

            CGPROGRAM  

            //使用vert函数和frag函数  

            #pragma vertex vert  

            #pragma fragment frag  

            #include "UnityCG.cginc"  

            fixed4 _OutlineCol;  

            float _OutlineFactor;  

            struct v2f  
            {  

                float4 pos : SV_POSITION;  
            };  

            v2f vert(appdata_full v)  
            {  

                v2f o;  
                //在vertex阶段，每个顶点按照法线的方向偏移一部分，不过这种会造成近大远小的透视问题  
                //v.vertex.xyz += v.normal * _OutlineFactor;  
                o.pos = UnityObjectToClipPos(v.vertex); 
                //将法线方向转换到视空间  
                float3 vnormal = mul((float3x3)UNITY_MATRIX_IT_MV, v.normal);  
                //将视空间法线xy坐标转化到投影空间，只有xy需要，z深度不需要了  
                float2 offset = TransformViewToProjection(vnormal.xy);  
                //在最终投影阶段输出进行偏移操作  
                o.pos.xy += offset * _OutlineFactor;  

                
                // float4 pos = mul(UNITY_MATRIX_MV, v.vertex); 
                // float3 normal = mul((float3x3)UNITY_MATRIX_IT_MV, v.normal);  
                // normal.z = -0.5;
                // pos = pos + float4(normalize(normal), 0) * _OutlineFactor;
                // o.pos = mul(UNITY_MATRIX_P, pos);

                return o;  
            }                

            fixed4 frag(v2f i) : SV_Target  
            {  
                //这个Pass直接输出描边颜色  
                return _OutlineCol;  
            }  
            ENDCG  
        }  

        Pass{
            Name "shadow"
            Tags { "Queue" = "Transparent-100" "RenderType" = "Transparent" }

            LOD 100
            Blend SrcAlpha OneMinusSrcAlpha            
            Cull Front  
            ZWrite Off     

            CGPROGRAM
            #include "Lighting.cginc"
            #pragma vertex vert
            #pragma fragment frag

            float _shadowOffset;
            float _shadowHeight;
            float4  _shadowColor;
            struct v2f
            {
                float4 pos : SV_POSITION;
            };

            v2f vert(appdata_base v)
            {
                v2f o;
                float4 worldPos = mul(unity_ObjectToWorld, v.vertex);
                
                //得到光源方向
                float3 lightDir = float3(worldPos.x,worldPos.y - _shadowOffset,worldPos.z);
                //float3 lightDir = normalize(UnityWorldSpaceLightDir(worldPos));
                //Y轴取反
                float PosY = _shadowHeight - worldPos.y;

                //得到缩放比,简单来说就是光照方向的倾斜度
                float rate = PosY / lightDir.y;
                //利用这个斜度得到X轴和Z轴
                float PosX = lightDir.x* rate + worldPos.x;
                float PosZ = lightDir.z* rate + worldPos.z;
                //得到最终的世界坐标
                float4 FinalworldPos = float4(PosX, _shadowHeight, PosZ, 1.0);
                //注意 因为我们用的是世界坐标了 所以变换矩阵是VP 不是常用的MVP了
                o.pos = mul(UNITY_MATRIX_VP, FinalworldPos);

                // float3 vnormal = mul((float3x3)UNITY_MATRIX_IT_MV, v.normal);  
                // //将视空间法线xy坐标转化到投影空间，只有xy需要，z深度不需要了  
                // float2 offset = TransformViewToProjection(vnormal.xy);  

                // o.pos.xy += offset * _shadowOffset;  

                return o;
            }

            float4 frag(v2f i) : SV_Target
            {
                //返回这个属性的颜色就行了
                return _shadowColor;
            }
            ENDCG
        }
        
        //正常着色的Pass  

        Pass  

        {  
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            // make fog work
            #pragma multi_compile_fog

            #include "UnityCG.cginc"

            struct appdata
            {
                float4 vertex : POSITION;
                float2 uv : TEXCOORD0;
            };

            struct v2f
            {
                float2 uv : TEXCOORD0;
                UNITY_FOG_COORDS(1)
                float4 vertex : SV_POSITION;
            };

            sampler2D _MainTex;
            float4 _MainTex_ST;

            v2f vert (appdata v)
            {
                v2f o;
                o.vertex = UnityObjectToClipPos(v.vertex);
                o.uv = TRANSFORM_TEX(v.uv, _MainTex);
                UNITY_TRANSFER_FOG(o,o.vertex);
                return o;
            }

            fixed4 frag (v2f i) : SV_Target
            {
                // sample the texture
                fixed4 col = tex2D(_MainTex, i.uv);
                // apply fog
                UNITY_APPLY_FOG(i.fogCoord, col);
                return col;
            }
            ENDCG
        }  
    }  
}
